
BAR

PURPOSE
Specifies if a bar will be drawn at the plot points of each trace on subsequent plots.

DESCRIPTION
DATAPLOT can draw a trace as a character or plot symbol at each point, as a connected line, as a spike from the point to a base, as a bar
from the plot point to a base, or as some combination of the above. The choice is determined by the BAR, SPIKE, CHARACTER, and
LINE commands. The switches for these commands work independently so that a plot point can be any combination of these (or even
none of them). The BAR switch is most commonly used to generate bar charts.

The following types of bar charts are commonly used in business and presentation graphics:

1. Standard bar charts (a bar is drawn from the data point to the X axis);

2. Grouped bar charts (bars are drawn for 2 or more groups of data);

3. Stacked (or divided) bar charts (the bar is divided into several intervals).

All of these formats are possible using DATAPLOT. At least one example of each of these formats is given in the PROGRAM section.
DATAPLOT provides commands to control the various aspects of the bar (see the RELATED COMMANDS section). Many of these
are also demonstrated with the sample programs.

Many analysts prefer that bar charts be drawn vertically rather than horizontally. This is straightforward to do and is demonstrated with
the sample programs.

SYNTAX
BAR <ON/OFF> <ON/OFF> <ON/OFF> etc.
where ON specifies that the trace is to be drawn with bars and OFF specifies that it is not. Up to 100 bar switches can be specified.

EXAMPLES
BAR ON OFF ON OFF
BAR OFF OFF ON
BAR ON ALL
BAR ALL ON
BAR

NOTE 1
An alternative to bar charts recommended by Bill Cleveland in “Elements of Graphing Data” is the dot chart. See the documentation for
the SPIKE command for an example of this.

NOTE 2
Although DATAPLOT has a BAR PLOT command, the recommended method for generating bar charts is to use the BAR switch with
the standard PLOT command.

NOTE 3
It is common with bar charts to want to use alphabetic labels. The TIC LABEL FORMAT, TIC LABEL CONTENT, and the TIC
OFFSET commands (documented in the Plot Control chapter) make this fairly straightforward. The sample program also demonstrates
their use.

NOTE 4
The BAR command with no arguments sets the bar type to blank for all traces. The BAR command with the word ALL before or after
the specified type assigns that bar type to all traces. Thus BAR ON ALL or BAR ALL ON plots bars for all traces.

DEFAULT
No bars are drawn (i.e., all OFF).

SYNONYMS
None

RELATED COMMANDS
PLOT = Generates a data or function plot.

CHARACTERS = Sets the character type for plot points.
LINES = Sets the line type for plot points.
SPIKES = Sets the on/off switches for plot spikes.
BAR BASE = Sets the base location for bars on plots
BAR FILL = Sets the on/off switches for plot bar fills.
BAR FILL COLOR = Sets the color for bar fills.
BAR DIMENSION = Sets the bar dimension to 2d or 3d.
BAR DIRECTION = Sets the bar direction to horizontal or vertical.
BAR PATTERN = Sets the bar fill pattern type.
BAR PATTERN COLOR = Sets the color for bar fill patterns.
BAR PATTERN LINE = Sets the line types for bar fill patterns.
BAR PATTERN SPACING = Sets the line spacing for bar fill patterns.
BAR PATTERN THICK = Sets the line thickness for bar fill patterns.
BAR BORDER COLOR = Sets the color for bar border lines.
BAR BORDER LINE = Sets the types for bar border lines.
BAR BORDER THICKNESS = Sets the line thickness for bar border lines.
BAR WIDTH = Sets the width for plot bars.

REFERENCE
“Elements of Graphing Data,” William S. Cleveland, Wadsworth Advanced Books and Software, 1985.

APPLICATIONS
Presentation Graphics

IMPLEMENTATION DATE
Pre-1987

PROGRAM 1
LET X = DATA 81 82 83 84 85
LET Y = DATA 2 5 9 15 28
LET Y2 = DATA 1 4 7 12 20
.
X1TIC MARK LABEL FORMAT ALPHA
X1TIC LABEL CONTENT 1981 1982 1983 1984 1985
XLIMITS 81 85; XTIC OFFSET 1 1
X1LABEL YEAR; MINOR X1TIC MARK NUMBER 0
Y1LABEL SALES (IN MILLIONS OF DOLLARS)
YLIMITS 0 30; MAJOR YTIC MARK NUMBER 4
MINOR YTIC MARK NUMBER 1; YTIC MARK SIZE 1.2
.
MULTIPLOT 3 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
.
TITLE BAR CHART WITH NO OPTIONS
LINE BLANK ALL
BAR ON ALL
PLOT Y X
.
BAR WIDTH .5 ALL
TITLE BAR CHART WITH USER DEFINED BAR WIDTH
PLOT Y X
.
BAR DIMENSION 3 ALL
TITLE BAR CHART WITH 3-DIMENSIONAL EFFECT
PLOT Y X
.
BAR FILL ONTS ALL
TITLE BAR CHART WITH 3-DIMENSIONAL EFFECT, FILLED
PLOT Y X
.
TITLE GROUP BAR CHART
LET X2 = X - 0.1; LET X3 = X + 0.1
BAR DIMENSION 2 ALL; BAR WIDTH 0.2 ALL
BAR FILL ON ALL; BAR FILL COLOR G25 G75
PLOT Y X2 AND
PLOT Y2 X3
.
TITLE DEMONSTRATE FILL PATTERNS
BAR FILL COLOR BLACK ALL
BAR PATTERN D1 D2
BAR PATTERN SPACING 2 2
BAR PATTERN THICKNESS 0.1 ALL
PLOT Y X2 AND
PLOT Y2 X3
.
THICKNESS 0.1; JUSTIFICATION LECE; HW 2 1
BOX FILL PATTERN D1; BOX FILL GAP 2
BOX 18 85 22 89
MOVE 24 87; TEXT GLC()ROUP 1
BOX FILL PATTERN D2
BOX 18 79 22 83
MOVE 24 81; TEXT GLC()ROUP 2
.
MULTIPLOT OFF

PROGRAM 2
LET CARTER = DATA 66 30 11 43 44 41 35 82 54 36
LET REAGAN = DATA 26 54 84 47 51 51 52 14 36 55
LET ANDERSON = DATA 6 12 4 8 3 6 10 3 7 8
LET X = DATA 1 2 3 5 6 7 8 10 11 12
LET X1 = X - 0.2; LET X2 = X; LET X3 = X + 0.2
.
. FOLLOWING SWITCHES ARE COMMON TO ALL OF THE FOLLOWING PLOTS
HORIZONTAL SWITCH ON
YTIC SIZE 1.2; YLIMITS 1 12; YTIC OFFSET 1 1; MINOR Y1TIC MARK NUMBER 0
Y1TIC MARK LABEL FORMAT ALPHA
Y1TIC LABEL CONTENT DEMOCRATS INDEPENDENTS REPUBLICANS SP() EAST ...

SOUTH MIDWEST WEST SP() BLACKS HISPANICS WHITES SP()
X1LABEL PERCENT; XTIC OFFSET 0 0; X1TIC MARK LABEL FORMAT DEFAULT
MINOR X1TIC MARK NUMBER DEFAULT; XLIMITS 0 100
MAJOR XTIC MARK NUMBER 11; MINOR XTIC MARK NUMBER 1
YTIC LABEL SIZE 1.6; HEIGHT 1.5
FRAME CORNER COORDINATES 20 20 90 90
.
. PLOT 1 (UPPER LEFT CORNER)
.
WINDOW CORNER COORDINATES 0 50 50 100
LINE BLANK BLANK BLANK
BAR ON ON ON
BAR BORDER THICKNESS 0.1 ALL
BAR WIDTH 0.2 ALL
BAR FILL ON ON OFF
BAR FILL COLOR BLACK G50 WHITE
TITLE VERTICAL GROUP BAR CHART
PLOT CARTER X1 AND
PLOT REAGAN X2 AND
PLOT ANDERSON X3
PRE-ERASE OFF
. CREATE THE LEGEND
JUSTIFICATION LECE
BOX FILL PATTERN ON; BOX FILL COLOR BLACK
BOX 75 85 78 88
MOVE 79 86.5; TEXT CARTER
BOX FILL COLOR G50
BOX 75 80 78 83
MOVE 79 81.5; TEXT REAGAN
BOX FILL PATTERN OFF
BOX 75 75 78 78
MOVE 79 76.5; TEXT ANDERSON
JUSTIFICATION LEFT
.
. PLOT 2 (UPPER RIGHT CORNER)
.
WINDOW CORNER COORDINATES 50 50 100 100
TITLE PYRAMID CHART
LINE BLANK BLANK SOLID
BAR WIDTH 0.5 ALL
BAR ON ON OFF
BAR FILL COLOR G25 G75
XLIMITS -100 100
LET CART2 = -CARTER
LET XJUNK = DATA 0.5 12.5

LET YJUNK = DATA 0 0
LEGEND 1 CARTER
LEGEND 1 COORDINATES 23 85
LEGEND 2 REAGAN
LEGEND 2 COORDINATES 87 85
LEGEND 2 JUSTIFICATION RIGHT
PLOT REAGAN VS X AND
PLOT CART2 VS X AND
PLOT YJUNK XJUNK
LEGEND 1 ; LEGEND 2 ; XLIMITS 0 100
.
. PLOT 3 (LOWER LEFT)
.
WINDOW CORNER COORDINATES 0 0 50 50
BAR BASE 0
BAR ON ALL
BAR FILL ON ALL
BAR FILL COLOR BLACK
TITLE DIVIDED BAR CHART
PLOT CARTER X
.
FRAME OFF; TIC MARKS OFF; TIC MARK LABELS OFF; TITLE
.
BAR BASE AUTOMATIC CARTER
BAR FILL COLOR G50 ALL
LET Y = REAGAN + CARTER
PLOT Y X X
BAR BASE AUTOMATIC Y
BAR FILL OFF ALL
LET Y2 = REAGAN + CARTER + ANDERSON
PLOT Y2 X X
.
FRAME ON; X1TIC MARKS ON; Y1TIC MARKS ON
X1TIC MARK LABELS ON; Y1TIC MARK LABELS ON
.
. PLOT 4 (LOWER RIGHT)
.
WINDOW CORNER COORDINATES 50 0 100 50
BAR FILL ON
BAR FILL COLOR BLACK
BAR BASE 0
FRAME CORNER COORDINATES 20 20 40 90
MAJOR XTIC MARK NUMBER 6; TITLE CARTER; X1LABEL
PLOT CARTER X
Y1TIC MARKS OFF; Y1TIC MARK LABELS OFF;
FRAME CORNER COORDINATES 45 20 65 90
TITLE REAGAN; X1LABEL PERCENT
PLOT REAGAN X
FRAME CORNER COORDINATES 70 20 90 90
ITLE ANDERSON; X1LABEL
PLOT ANDERSON X

